

Risoluzione di un Sistema Elettrico Trifase Simmetrico e Squilibrato*

Davide Tambuchi

8 febbraio 2004

1 La Teoria

Nel presente lavoro, esamineremo un programma in FORTRAN77 utilizzato per il calcolo delle correnti e delle tensioni in un carico trifase *squilibrato* collegato a *stella*, in *assenza di neutro*, ed alimentato da una *terna simmetrica* di tensioni.

Come noto, per determinare le tensioni e le correnti sul carico, occorre innanzitutto determinare lo *spostamento del centro stella* $\dot{E}_{00'}$.




Figura 1: Sistema trifase simmetrico e squilibrato senza neutro

Con riferimento alla figure 1 e 2, si ha:

$$\dot{E}_{00'} = \frac{\frac{\dot{E}_1}{Z_1} + \frac{\dot{E}_2}{Z_2} + \frac{\dot{E}_3}{Z_3}}{\frac{1}{Z_1} + \frac{1}{Z_2} + \frac{1}{Z_3}} \quad (1)$$

Determinato lo spostamento del centro stella, é possibile calcolare le tensioni sul carico:

$$\dot{E}_1 = \dot{E}_{10} - \dot{E}_{00'} \quad (2)$$

*Questo documento può essere liberamente distribuito, purché senza modifiche, integralmente e senza scopo di lucro o altri scopi commerciali. Ogni cura é stata posta nella stesura del documento. Tuttavia l'Autore non può assumersi alcuna responsabilità derivante dall'utilizzo della stessa.




Figura 2: Diagramma vettoriale delle tensioni

$$\dot{E}_2 = \dot{E}_{20} - \dot{E}_{00'} \quad (3)$$

$$\dot{E}_3 = \dot{E}_{30} - \dot{E}_{00'} \quad (4)$$

A questo punto, é immediato determinare le correnti:

$$\dot{I}_1 = \frac{\dot{E}_1}{\dot{Z}_1} \quad (5)$$

$$\dot{I}_2 = \frac{\dot{E}_2}{\dot{Z}_2} \quad (6)$$

$$\dot{I}_3 = \frac{\dot{E}_3}{\dot{Z}_3} \quad (7)$$

2 Il Programma

Il programma richiede in ingresso il valore efficace della tensione *stellata* che alimenta il circuito, nonché i valori complessi delle impedenze costituenti il carico. In uscita vengono visualizzati i valori delle tensioni e delle correnti relative al carico, sia come numeri complessi, e sia in modulo e fase. Si é scelto il linguaggio FORTRAN77 in quanto possiede una implementazione dei numeri complessi che rende agevole il calcolo. A tal proposito si utilizzano le funzioni ABS e ATAN2; la prima determina il modulo di un numero complesso, e la seconda l'arcotangente di un angolo, note le parti reali ed immaginaria, determinando correttamente l'angolo (in *radianti*), che risulta compreso tra $-\pi$ e $+\pi$. Per comodità, tale angolo viene convertito in gradi.

```
PROGRAM SQUIL
```

```
* QUESTO PROGRAMMA DETERMINA LE TENSIONI E LE CORRENTI IN UN CARICO TRIFASE
* SQUILIBRATO A STELLA, ALIMENTATO DA UNA TERNA SIMMETRICA DI TENSIONI.
```

```
REAL GRAD
PARAMETER (GRAD = 180.0 / 3.14159265)
```

```

*           FATTODE DI CONVERSIONE DA RADIANTI A GRADI

REAL ANGOLO
PARAMETER (ANGOLO = 120.0 / GRAD)
*           ANGOLO TRA LE FASI DI ALIMENTAZIONE IN RADIANTI

REAL E
*           VALORE EFFICACE DELLA TENSIONE (STELLATA) DI ALIMENTAZIONE

COMPLEX E10, E20, E30
*           VALORI COMPLESSI DELLE TENSIONI DI ALIMENTAZIONE

COMPLEX Z1, Z2, Z3
*           IMPEDENZE COSTITUENTI IL CARICO

COMPLEX I1, I2, I3
*           CORRENTI NELLE TRE FASI

COMPLEX E1, E2, E3
*           TENSIONI SULLE TRE IMPEDENZE DI CARICO

COMPLEX E00
*           SPOSTAMENTO DEL CENTRO STELLA

*           INTRODUZIONE DELLA TENSIONE DI ALIMENTAZIONE E DELLE
*           IMPEDENZE DI CARICO

WRITE (UNIT=*, FMT=*)
+ ' TENSIONE DI ALIMENTAZIONE STELLATA (REALE) : '
READ (UNIT=*, FMT=*) E

WRITE (UNIT=*, FMT=*) ' IMPEDENZA Z1 (RE, IM) : '
READ (*,*) Z1

WRITE (UNIT=*, FMT=*) ' IMPEDENZA Z2 (RE, IM) : '
READ (*,*) Z2

WRITE (UNIT=*, FMT=*) ' IMPEDENZA Z3 (RE, IM) : '
READ (*,*) Z3

*           CALCOLO DELLE TENSIONI (COMPLESSE) DI ALIMENTAZIONE

E10 = CMLPX(E, 0.0)
E20 = CMLPX(E, 0.0) * CMLPX(COS(ANGOLO), -SIN(ANGOLO))
E30 = CMLPX(E, 0.0) * CMLPX(COS(ANGOLO), SIN(ANGOLO))

*           CALCOLO DELLO SPOSTAMENTO DEL CENTRO STELLA

E00 = (E10/Z1 + E20/Z2 + E30/Z3)/(1/Z1 + 1/Z2 + 1/Z3)

*           CALCOLO DELLE TENSIONI STELLATE SUL CARICO

E1 = E10 - E00
E2 = E20 - E00
E3 = E30 - E00

*           CALCOLO DELLE CORRENTI

I1 = E1 / Z1
I2 = E2 / Z2
I3 = E3 / Z3

```

```

*          VISUALIZZAZIONE DEL RISULTATO:
*          MEDIANTE LE FUNZIONI ABS E ATAN2 IL RISULTATO
*          VIENE VISUALIZZATO ANCHE IN MODULO E FASE

          WRITE (UNIT=*,FMT=170)
*          RIGA DI RIFERIMENTO

          WRITE (UNIT=*, FMT=100) E00, ABS(E00),
+          GRAD * ATAN2(AIMAG(E00), REAL(E00))
          WRITE (UNIT=*, FMT=110) E1, ABS(E1),
+          GRAD * ATAN2(AIMAG(E1), REAL(E1))
          WRITE (UNIT=*, FMT=120) E2, ABS(E2),
+          GRAD * ATAN2(AIMAG(E2), REAL(E2))
          WRITE (UNIT=*, FMT=130) E3, ABS(E3),
+          GRAD * ATAN2(AIMAG(E3), REAL(E3))
          WRITE (UNIT=*, FMT=140) I1, ABS(I1),
+          GRAD * ATAN2(AIMAG(I1), REAL(I1))
          WRITE (UNIT=*, FMT=150) I2, ABS(I2),
+          GRAD * ATAN2(AIMAG(I2), REAL(I2))
          WRITE (UNIT=*, FMT=160) I3, ABS(I3),
+          GRAD * ATAN2(AIMAG(I3), REAL(I3))

          STOP

170 FORMAT (32X, 'REALE', 7X, 'IMMAG', 7X, 'MODULO', 7X, 'FASE')
100 FORMAT (' SPOST. CENTRO STELLA E00 = ',
+          4(SP, F10.3, 2X))
110 FORMAT (' TENSIONE E1           = ',
+          4(SP, F10.3, 2X))
120 FORMAT (' TENSIONE E2           = ',
+          4(SP, F10.3, 2X))
130 FORMAT (' TENSIONE E3           = ',
+          4(SP, F10.3, 2X))
140 FORMAT (' CORRENTE I1            = ',
+          4(SP, F10.3, 2X))
150 FORMAT (' CORRENTE I2            = ',
+          4(SP, F10.3, 2X))
160 FORMAT (' CORRENTE I3            = ',
+          4(SP, F10.3, 2X))

          END

```

Riferimenti bibliografici

- [1] M. Pezzi. *Elettrotecnica Generale*, Zanichelli, Bologna (1978).
- [2] R. Kumar. *Programming with FORTRAN77*, Tata McGraw-Hill Publishing Company Limited, New Delhi, (1986).

Typeset by L^AT_EX 2_ε under LINUX

e-mail: davide.tambuchi@tin.it